unittest
{
	assert( [ "1 0" ].parse.expand.solve == "0 0 1 1" );
	assert( [ "1 1" ].parse.expand.solve == "-1" );
	assert( [ "5 58" ].parse.expand.solve == "-1" );

	assert( [ "2 1" ].parse.expand.solve == "0 2 1 2 0 3 1 3" );		// 追加ケース
	assert( [ "2 2" ].parse.expand.solve == "0 1 2 1 0 3 2 3" );		// 追加ケース
	assert( [ "2 3" ].parse.expand.solve == "0 1 3 1 0 2 3 2" );		// 追加ケース
	assert( [ "2 4" ].parse.expand.solve == "-1" );						// 追加ケース
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto mk = input.front.split.to!( long[] );
	return tuple( mk[ 0 ], mk[ 1 ] );
}

auto solve( long m, long k )
{
	auto n = 2L ^^ ( m + 1 );
	auto h = n / 2;
	auto qa = n / 4;
	auto qb = h + qa;
	
	// kが0の場合は小さい順に並べるだけでOK
	if( k == 0 ) return n.iota.map!( ( a ) => ( a / 2 ).to!( string ) ).join( " " );
	
	// kが0より大きい場合はk以外の数全てのxorがkにならないと無理
	if( h <= k || reduce!( ( a, b ) => a ^ b )( k, h.iota ) != k ) return "-1";
	
	// kはk以外の数全てを一つだけ挟むように並べる
	// k以外の数はkを一つだけ挟んで、それ以外は挟まないか二つ挟んで相殺するように並べる
	auto anss = new long[ n ];
	anss[ qa ] = k;
	anss[ qb ] = k;
	auto a = 0L;
	for( auto i = 1L; i < qa; i++ )
	{
		a++;
		a += ( a == k );
		anss[ qa + i ] = a;
		anss[ qa - i ] = a;
		a++;
		a += ( a == k );
		anss[ qb + i ] = a;
		anss[ qb - i ] = a;
	}
	return anss.to!( string[] ).join( " " );
}
